

Head-Poke

```
100 MODE 2
110 FOR adr=&B0C7 TO &B0DA
120 READ a$:POKE adr,VAL("&" + a$)
130 NEXT
140 DATA E5,D5,C5,21,DB,B0,11,68
150 DATA B8,01,24,00,ED,B0,C1,D1
160 DATA E1,C3,4A,13
170 :
180 : '36 Stellen stehen zur Verfuegung
190 d$="CPC-Magazin: Tips&Tricks...wunde
rbar"
200 FOR a=&B0DB TO &B0FE
210     POKE a,ASC(MID$(d$,a-&B0DA,1))
220 NEXT
230 :
240 POKE &BDD4,&C7:POKE &BDD5,&B0
250 :
260 PRINT"Bitte Cassette wechseln."
270 PRINT
280 SAVE"HEADER TEST NR.1",A
290 PRINT
300 :
310 PRINT"Output-Header: &B84C -> &B88B"
320 PRINT
330 FOR a=&B84C TO &B88B STEP 16
340     PRINT HEX$(a,4)": ";
350     FOR b=0 TO 15
360         c=PEEK(a+b)
370         IF c>=32 AND c<123 THEN PRINT "
"CHR$(c);" "; ELSE PRINT HEX$(c,2);" ";
380     NEXT
390     PRINT
400 NEXT
410 :
420 PRINT
430 PRINT"Bitte Cassette zurueckspulen."
440 PRINT
450 CHAIN"HEADER TEST NR.1",460
460 PRINT
470 :
480 PRINT"Input-Header : &B88C -> &B8CB"
490 PRINT
500 FOR a=&B88C TO &B8CB STEP 16
510     PRINT HEX$(a,4)": ";
520     FOR b=0 TO 15
530         c=PEEK(a+b)
540         IF c>=32 AND c<123 THEN PRINT "
"CHR$(c);" "; ELSE PRINT HEX$(c,2);" ";
550     NEXT
560     PRINT
570 NEXT
580 :
590 POKE &BDD4,&4A:POKE &BDD5,&13
```

Source-Code

```
100 'org &b0c7
110 'push hl
120 'push de
130 'push bc
140 'ld hl,&b0db
150 'ld de,&b868
160 'ld bc,36
170 'ldir
180 'pop bc
190 'pop de
200 'pop hl
210 'jp &134a
220 'end
```